
 
 



 
 
 
 



program mislog 
version 14.2 
args todo eta mu return 
if `todo' == -1 { 
 /* Set global macros for output */ 
 global SGLM_lt "misclassified log" 
 global SGLM_lf "log( (p-0.127)/0.681)" 
 exit 
} 
if `todo' == 0 { 
 /* set eta =g(mu) */ 
 generate double `eta' = log( ((`mu' - 0.127)/0.681) ) 
 exit 
} 
if `todo' == 1 { 
 /* set g^{-1}(eta) */ 
 generate double `mu' = exp(`eta')*0.681 + 0.127 
 exit 
} 
if `todo' == 2 { 
 /* set return = dmu/deta */ 
    local d2 = 0.681*exp(`eta') 
 generate double `return' = 0.681*exp(`eta') 
exit 
} 
if `todo' == 3 { 
 /* set return =d2mu/dmu2  */ 
 generate double `return' = 0.681*exp(`eta') 
 exit 
} 
display as error "Unknown call to glm link function" 
exit 198 
end 
 
 
 
program mislogit 
version 14.2 
args todo eta mu return 
if `todo' == -1 { 
 /* Set global macros for output */ 
 global SGLM_lt "logit with missclassification" 
 global SGLM_lf "logit(((p - 0.127)/0.681))" 
 exit 



} 
if `todo' == 0 { 
 /* set eta =g(mu) */ 
 generate double `eta' = logit( ((`mu' - 0.127)/0.681) ) 
 exit 
} 
if `todo' == 1 { 
 /* set g^{-1}(eta) */ 
 generate double `mu' = (invlogit(`eta')*0.681) + 0.127 
 exit 
} 
if `todo' == 2 { 
 /* set return = dmu/deta */ 
 generate double `return' = 0.681*invlogit(`eta')*(1-invlogit(`eta')) 
exit 
} 
if `todo' == 3 { 
 /* set return =d2mu/dmu2  */ 
 generate double `return' = 0.681*( exp(`eta')*(1-exp(`eta')) )/( (1+exp(`eta'))^3 ) 
 exit 
} 
display as error "Unknown call to glm link function" 
exit 198 
end 
 
 
 


